

# **Algunas Funcionalidades Especiales de SPICE3**

**Dr. José Ernesto Rayas-Sánchez**

1

## **Versiones de SPICE Estándar**

---

- La mayoría de los actuales simuladores tipo SPICE comerciales están basados en una de las siguientes dos versiones estándar del lenguaje:
  - SPICE3F4 (o simplemente SPICE 3)
  - SPICE2G6 (o simplemente SPICE 2)
- WinSpice es compatible con ambas versiones
- SPICE 3 tiene mayores capacidades de programabilidad

## Bloque de Control en WinSpice

- SPICE 3 (disponible en WinSpice) permite definir todos los comandos dentro de un bloque de control
- El bloque de control comienza con `.CONTROL` y termina con `.ENDC`
- Dentro del bloque de control se pueden definir comandos para ejecutar análisis (AC, TRAN, etc.), se pueden definir expresiones (`real(v(5))`, etc.), se pueden definir comandos para graficar (PLOT), y se pueden definir pequeños programas para simulaciones paramétricas (ciclos, etc.)

## Variables de Salida en CA

- Los siguientes sufijos son aplicables sólo dentro de los comandos PLOT, PRINT, WRITE, etc.

<i>V</i>	magnitud (equivalente a <i>VM</i> )
<i>VR</i>	parte real
<i>VI</i>	parte imaginaria
<i>VM</i>	magnitud
<i>VP</i>	fase
<i>VDB</i>	decibelios = $20 \log_{10}(\text{magnitud})$

(para corrientes, sustituir *V* por *I*)


- **No son aplicables para asignar variables en un bloque de control**

## Análisis y Graficación en SPICE 2 – Ej. 1

- Voltajes en AC: ( $v(out)=vm(out)$ )

```
.AC DEC 50 10Hz 900MEGHZ
```

```
.plot ac v(out) vm(out) vr(out) vi(out)
```



Dr. J. E. Rayas-Sánchez


5

## Análisis y Graficación en SPICE 2 – Ej. 2

- Corrientes en AC: ( $i(Vs)=im(Vs)$ )

```
.AC DEC 50 10Hz 900MEGHZ
```

```
.plot ac i(Vs) im(Vs) ir(Vs) ii(Vs)
```



Dr. J. E. Rayas-Sánchez

6



## Análisis y Graficación en SPICE 2 – Ej. 3

- Ángulos de fase de voltajes y corrientes en AC:

```
.AC DEC 50 10Hz 900MEGHZ
```

```
.plot ac vp(out)
```

```
.plot ac ip(Vs)
```



Dr. J. E. Rayas-Sánchez

7



## Análisis y Graficación en SPICE 2 – Ej. 4

- Voltajes y corrientes transitorios:

```
.TRAN 10us 5ms
```

```
.plot tran v(in) v(out)
```

```
.plot tran i(Vs)
```




Dr. J. E. Rayas-Sánchez

8

## Análisis y Graficación en SPICE 3 – Ej. 1

- Voltajes en AC: ( $v(out)=vm(out)=Vout\_mag$ )

```
.CONTROL
AC DEC 50 10Hz 900MEGHZ
Vout_mag = mag(v(out))
Vout_real = real(v(out))
Vout_imag = im(v(out))
plot Vout_mag Vout_real Vout_imag
plot v(out) vm(out) vr(out) vi(out)
.ENDC
```




Dr. J. E. Rayas-Sánchez

9

## Análisis y Graficación en SPICE 3 – Ej. 2

- Corrientes en AC: ( $i(Vs)=im(Vs)=Iin\_mag$ )

```
.CONTROL
AC DEC 50 10Hz 900MEGHZ
Iin_mag = mag(i(Vs))
Iin_real = real(i(Vs))
Iin_imag = im(i(Vs))
plot Iin_mag Iin_real Iin_imag
plot i(Vs) im(Vs) ir(Vs) ii(Vs)
.ENDC
```







Dr. J. E. Rayas-Sá

10

## Análisis y Graficación en SPICE 3 – Ej. 3

- Voltajes y corrientes transitorios:

```
.CONTROL
TRAN 10us 5ms
Vin = v(in)
Vout = v(out)
Iin = i(Vs)
plot Vin Vout
plot Iin
plot v(in) v(out)
plot i(Vs)
.ENDC
```





Dr. J. E. Rayas-Sánchez

11

## Análisis y Graficación en SPICE 3 – Ej. 4

- Magnitudes de ganancias e impedancias:

```
.control
destroy all
AC DEC 10 10Hz 900MEGHZ
mAv = mag(v(out)/v(in))
mZin = mag(v(in)/i(Vs))
plot mAv
plot mZin
.endc
```




Dr. J. E. Rayas-S

2

## Análisis y Graficación en SPICE 3 – Ej. 5a

- Ángulos de fase de ganancias e impedancias:

```
.control
destroy all
set units = radians
AC DEC 10 10Hz 900MEGHZ
pAv = phase(v(out)/v(in))
pZin = phase(v(in)/-i(Vs))
plot pAv pZin
.endc
```




Dr. J. E. Rayas-Sánchez

## Análisis y Graficación en SPICE 3 – Ej. 5b

- Ángulos de fase de ganancias e impedancias:

```
.control
destroy all
set units = degrees
AC DEC 10 10Hz 900MEGHZ
pAv = phase(v(out)/v(in))
pZin = phase(v(in)/-i(Vs))
plot pAv pZin
.endc
```




Dr. J. E. Rayas-Sánchez

## Análisis y Graficación en SPICE 3 – Ej. 6

- Potencias instantáneas:

```
.CONTROL  
TRAN 10us 5ms  
Pin = -v(in)*i(Vs)  
Pout = v(out)*v(out)/1K  
plot Pin Pout  
.ENDC
```




Dr. J. E. Rayas-Sánchez

15

## Gráficas de Ruido en SPICE 2

```
.options TEMP = 27  
.NOISE v(out) Vs DEC 50 10Hz 900MEGHZ  
.plot noise inoise onoise
```




Dr. J. E. Rayas-Sánchez

16



## Gráficas de Ruido en SPICE 3

```
.control
options TEMP = 27
NOISE v(out) Vs DEC 50 10Hz 900MEGHz
plot sqrt(inoise) sqrt(onoise)
.endc
```



Dr. J. E. Rayas-Sánchez

17

## Parámetros de Modelos y Componentes (SPICE 3)

- Parámetros en P. de O. de modelos y componentes:

```
.CONTROL
OP
* DC Voltages, Currents, and Powers:
VB = v(b); VE = v(e); VC = v(c); VCE = VC-VE; Vcc = v(vp)
IVcc = -i(Vcc); PVcc = Vcc*IVcc
* Instance Parameters:
Q1VBE = @Q1[vbe]; Q1gm = @Q1[gm]
Q1cpi = @Q1[cpi]; Q1cmu = @Q1[cmu]
* Model Parameters:
2N2222BF = @Q2N2222[BF]; 2N2222VAF = @Q2N2222[VAF]
print VB VE VC VCE Vcc IVcc PVcc
print Q1VBE Q1gm Q1cpi Q1cmu 2N2222BF 2N2222VAF
.ENDC
```

(semicolon is used only for  
a more compact display)

Dr. J. E. Rayas-Sánchez

18

## Parámetros de Modelos y Componentes (SPICE 3)

- Resultados:

```

WinSpice v5.3.5
File Edit Settings Help
TEMP=27 deg C
DC Operating Point ... 100%
vb = 6.878448e+00
ve = 6.144428e+00
vc = 1.019021e+01
vce = 4.045785e+00
vcc = 1.200000e+01
ivcc = 7.669245e-03
pvcc = 9.203094e-02
q1vbe = 7.340203e-01
q1gm = 2.578140e-01
q1cpi = 1.616684e-10
q1cmu = 8.301793e-12
2n222bf = 2.175000e+02
2n222vaf = 1.315000e+02
WinSpice 13 ->
    
```

## Parámetros de Modelos y Componentes (SPICE 3)

- Parámetros en P. de O. de modelos y componentes:

```

.CONTROL
OP
* DC Voltages, Currents, and Powers:
VB = v(b); VE = v(e); VC = v(c); VCE = VC-VE; Vcc = v(vp)
IVcc = -i(Vcc); PVcc = Vcc*IVcc
* Instance Parameters:
Q1VBE = @Q1[vbe]; Q1gm = @Q1[gm]
Q1cpi = @Q1[cpi]; Q1cmu = @Q1[cmu]
* Model Parameters:
2N2222BF = @Q2N2222[BF]; 2N2222VAF = @Q2N2222[VAF]
write results_1_op.csv VB VE VC VCE Vcc IVcc PVcc
write results_2_op.csv Q1VBE Q1gm Q1cpi Q1cmu
write results_3_op.csv 2N2222BF 2N2222VAF
.ENDC
    
```

(semicolon is used only for a more compact display)

## Parámetros de Modelos y Componentes (SPICE 3)

- Resultados:

The image shows three overlapping Excel spreadsheets, each displaying the results of a SPICE simulation for a different component. Each spreadsheet has a header row with labels and a data row with numerical values.

	A	B	C	D	E	F	G	H
1	in	vb	ve	vc	vce	vcc	ivcc	pvcc
2	0.00E+00	6.88E+00	6.14E+00	1.02E+01	4.05E+00	1.20E+01	7.67E-03	9.20E-02

	A	B	C	D	E	F	G	H
1	in	q1vbe	q1gm	q1cpi	q1cmu			
2	0.00E+00	7.34E-01	2.58E-01	1.62E-10	8.30E-12			

	A	B	C	D	E	F	G	H
1	in	2n2222bf	2n2222vaf					
2	0.00E+00	2.18E+02	1.32E+02					

Dr. J. E. Rayas-Sánchez

21

## Lenguaje de Interpretación Interactiva (SPICE 3)


- Simulación variando los elementos del circuito:

```
.CONTROL
destroy all
let count = 0
while count < 5
    alter RS = 85 + 85*count
    AC DEC 10 10 900MEGHZ
    TRAN 10E-6 5E-3
    let count = count + 1
end
plot db(ac1.v(out)) db(ac2.v(out)) db(ac3.v(out))
+ db(ac4.v(out)) db(ac5.v(out))
plot tran1.v(out) tran2.v(out) tran3.v(out)
+ tran4.v(out) tran5.v(out)
.ENDC
```

Dr. J. E. Rayas-Sánchez

22


## Lenguaje de Interpretación Interactiva (cont)



Dr. J. E. Rayas-Sánchez

23

## Lenguaje de Interpretación Interactiva (cont)



Dr. J. E. Rayas-Sánchez

24